QT += widgets printsupport

TEMPLATE = lib
DEFINES += BTWIDGET_LIBRARY QCUSTOMPLOT_COMPILE_LIBRARY

CONFIG += c++11 debug_and_release build_all

#load(qt_module)
CONFIG += build_btwidget_lib

static {
  CONFIG += static
} else {
  CONFIG += shared
}

# The following define makes your compiler emit warnings if you use
# any Qt feature that has been marked deprecated (the exact warnings
# depend on your compiler). Please consult the documentation of the
# deprecated API in order to know how to port your code away from it.
DEFINES += QT_DEPRECATED_WARNINGS

# You can also make your code fail to compile if it uses deprecated APIs.
# In order to do so, uncomment the following line.
# You can also select to disable deprecated APIs only up to a certain version of Qt.
#DEFINES += QT_DISABLE_DEPRECATED_BEFORE=0x060000    # disables all the APIs deprecated before Qt 6.0.0

TARGET = BtWidget
VERSION = 1.0.0

include(BtWidget.pri)
include(BtLayout.pri)

HEADERS += \
    BtWidget_global.h \
    btwidget.h

# Default rules for deployment.
unix {
    target.path = /usr/lib
}
!isEmpty(target.path): INSTALLS += target


CONFIG(debug, debug|release) {
compiled = debug
}
CONFIG(release, debug|release) {
compiled = release
}

MOC_DIR = $$compiled/.moc
UI_DIR = $$compiled/.ui
OBJECTS_DIR = $$compiled/.obj

CONFIG(debug, debug|release) {
  TARGET = $$join(TARGET,,,d) # if compiling in debug mode, append a "d" to the library name
  QMAKE_TARGET_PRODUCT = "BtWidget (debug mode)"
  QMAKE_TARGET_DESCRIPTION = "BtWidget library for Qt5 (debug mode)"
} else {
  QMAKE_TARGET_PRODUCT = "BtWidget"
  QMAKE_TARGET_DESCRIPTION = "BtWidget library for Qt5"
}
QMAKE_TARGET_COMPANY = "www.mr-xbt.cn"
QMAKE_TARGET_COPYRIGHT = "Copyright (C) by XiaoBotao"

# 复制动态库和  头文件
DEST_DD=E:\workspace\Qt\Deploy\BtWidget
target.path = $$DEST_DD\lib
lib_headers.path = $$DEST_DD\include
lib_headers.files = $$HEADERS
INSTALLS += target lib_headers
